GAMS program code for the DEA rail model  <insert number provided by MAPS> or ISSN as appropriate
**********************************************************************************                                                                               *

* The purpose of this program is to solve the DEA models for producing          *

* estimates of the DEA scores and related variables used in the                 *

* Productivity Commission (1999) reports Progress in Rail Reform; and           *
* Assessment of the Performance of Australian Railways, 1990 to 1998.           *                                       *                                                                               *

* The program code is written using the GAMS syntax. Details of the programming *

* language and software licensing can be found in Brooke, A., D. Kendrick,      *

* A. Meeraus, and R. Raman (1998) GAMS: A User’s Guide. GAMS Development        *

* Corporation, Washington DC, USA.                                              *

*                                                                               *

* The program performs the following functions:                                 *

* (i)   reads data from a text file;                                            *

* (ii)  calculates DEA variables (scores, slacks, and weights) using various    *

*       model specifications;                                                   *

* (iii) stores solutions in a useable format.                                   *

*                                                                               *

* The program is somewhat generic in that it can be used to implement a number  *

* of variations from the basic DEA problem. Various options / versions of the   *

* model can be chosen by commenting out a program statement or setting          *

* the related parameter to a particular value. The example presented estimates  *

* model 5 results as reported in the reports mentioned above.                   *

*                                                                               * 

* A number of authors contributed to the development of this program, including *

* Alexander Strezlecki, Chris Chan, Dr Yimin Zhou and Tendai Gregan.            *

* Dr John Salerian offered useful advice for setting up the core program        *

* structure.                                                                    *

*                                                                               *

*********************************************************************************

OPTION DECIMALS = 3;

OPTION LP       = MINOS5;

OPTION LIMCOL   = 0;

OPTION LIMROW   = 0;

OPTION SOLPRINT = OFF;

$OFFSYMLIST OFFSYMXREF

*  Logical file for storing the model solution in csv format  

FILE DEARESULT / solution.csv /;

DEARESULT.ps = 130;

DEARESULT.pw = 255;

DEARESULT.pc = 6;

SETS

i Individual and national railway systems 

   / TAA  Trans-Adelaide

     AN   AN and NRC combined 

     NSW  SRA of the New South Wales

     PTC  Public Transport Corporation of Victoria

     QRL  Queensland Rail

     WSR  Westrail

     CL1  Canada - Class I

     CL23 Canada - Classes II and III

     CN   Canadian National

     CP   Canadian Pacific

     CR   Consolidated Rail Corporation - United States

     CSX  CSX Transportation - United States

     GTW  Grand Trunk Western Inc - United States

     IC   Illinois Central Railroad Company - United States

     KCS  Kansas City Southern Corporation - United States

     NS   Norfolk Southern Corporation - United States

     SOO  SOO Line Railroad Company - United States

     UP   Union Pacific Railroad Company - United States

     BNSF Burlington Northern Santa Fe Railway Company - United States 

     USA  United States

     AUS  Australia

     CAN  Canada

     FRA  France

     GBR  Great Britain

     AST  Austria

     BEL  Belgium

     DEN  Denmark

     FIN  Finland

     GER  Germany

     IRE  Ireland

     ITA  Italy

     JAP  Japan

     LUX  Luxemburg

     NET  Netherlands

     NOR  Norway

     NZL  New Zealand

     POR  Portugal

     SPA  Spain

     SWE  Sweden

     SWI  Switzerland

     UK   United Kingdom

     RSA  Republic of South Africa /

********************************************************************************** The ‘coverage’ statement defines the sample coverage by selecting railways    *

* from the sample list above for inclusion in the model.                        *

* For example, the statement below is used in the context of model 5 to cover   *

* the national railway systems of the listed countries .                        *

*********************************************************************************

coverage(i) / USA,AUS,CAN,FRA,GBR,AST,BEL,DEN,FIN,GER,IRE,ITA,JAP,LUX,

              NET,NOR,NZL,POR,SPA,SWE,SWI,UK,RSA /

 y             Year                             / Y70 * Y98 /

 iter1         Loop counter                     / 1 * 50 /

 iy(i,y)       Set of observations that contain unknown data 

 window(y)     Window to be declared below; facility not used in the rail study

 sample(i,y)   Samples included in each DEA calculation 

 catgroup(i,y) Candidates of peer members included in each DEA calculation 

 iovar         Types of inputs, outputs, and environmental factors  

               / NTKMF         Net tonne kilometre

                 PASSKMS       Passenger kilometre

                 EMPLOF        Number of freight workers

                 TEMPS         Number of employee

                 LOCOS         Number of locomotives in service

                 LOCOF         Number of freight locomotives

                 WAGONF        Number of wagons 

                 PASSCAS       Number of passenger cars

                 TRACKMS       Track kilometre

                 HORSE         Locomotive power in kilowatt

                 DENSITY       Train density

                /

 cv(iovar)   Categorical data in the data set; not used in the rail study

 m(iovar)    Output data in the data set; model 5 example:    / NTKMF, PASSKMS /

 n(iovar)    Input data in the data set; model 5 example:

                                      / TEMPS, LOCOS, WAGONF, PASSCAS, TRACKMS /

 mm(m)       Output data used in the model; model 5 example:  / NTKMF, PASSKMS /

 nn(n)       Input data used in the model; model 5 example:

                                      / TEMPS, LOCOS, WAGONF, PASSCAS, TRACKMS /

 cnn(nn)     Discretionary inputs; model 5 example:

                                                / TEMPS, LOCOS, WAGONF, PASSCAS/

 ucnn(nn)    Non-discretionary inputs; model 5 example:              / TRACKMS /

;

ALIAS (i,ip)   ; 

ALIAS (y,yp)   ;

ALIAS (i,ipp)  ; 

ALIAS (y,ypp)  ;

ALIAS (i,icv)  ;

ALIAS (y,ycv)  ;

*  To read input and output data from the text file ‘datauic.prn’:

TABLE IODATA(i,y,iovar)     Annual outputs and inputs by system

$include datauic.prn

;

PARAMETERS

  DMUY(i,y,m)            Annual outputs by system

  DMUX(i,y,n)            Annual inputs by system

  DMUCV(i,y)             Categorical variables by system

  TINY                   Tiny number 

  ITER2                  Loop counter  

  ITER3                  Loop counter  

  WINWIDTH               Window width 

  YEARNO                 Number of years covered in the study period 

  ITERNO                 Number of windows 

  ZERO(i,y)              Detector for missing data

  DMUXX(i,y,n)           Inputs of comparators 

  DMUYY(i,y,m)           Outputs of comparators 

  DMUINP(n)              Inputs of the DMU being evaluated

  DMUOUT(m)              Outputs of the DMU being evaluated

  PHIHAT(i,y,*)          Efficiency score

  PEERHAT(*,i,y,ip,yp)   Weighting multipliers

  OUTSLACK(*,i,y,m)      Slack outputs, in percentage terms  

  INPEXCES(*,i,y,n)      Excess inputs, in percentage terms 

;

* To define input, output and categorical data: 

DMUX(i,y,n) = IODATA(i,y,n);

DMUY(i,y,m) = IODATA(i,y,m);

DMUCV(i,y)  = IODATA(i,y,"DENSITY");

TINY        = 0.00001;

* To detect and exclude observations with missing data:

ZERO(i,y)   = TINY * PROD(nn, DMUX(i,y,nn)) * SUM(mm, DMUY(i,y,mm))

              * TINY * DMUCV(i,y) ;

iy(i,y)                 = YES; 

iy(i,y) $ (ZERO(i,y)=0) = NO ;

*********************************************************************************

* Categorical data can be included in the model using the method described in   *

* Charnes, A., W. Cooper, A.Y. Lewin and L.S. Seiford (1994, Chapter 3) Data    *
* Envelopment Analysis: Theory, Methodology and Applications, Kluwer Academic   *

* Publishers, Massachusetts, USA.                                               *

*                                                                               *

* This model extension is not used in the rail study, and is disabled in the    *

* program code by setting DMUCV to 1.                                           *

*********************************************************************************

DMUCV(i,y)  = 1 ;

*********************************************************************************

* The technique of window analysis in DEA is described in Charnes et.al. ibid.  *

* This model extension is not used in the rail study, and is disabled in the    *

* program code by setting WINWIDTH to be equal to YEARNO.                       *

*********************************************************************************

WINWIDTH    = 29;

YEARNO      = 29;

ITERNO      = YEARNO - WINWIDTH + 1;

VARIABLES

  PHI           Efficiency score 

  LAMBDA(i,y)   Weighting multiplier 

  Z             Objective function value 

;

POSITIVE VARIABLES LAMBDA ; 

FREE     VARIABLES PHI Z ;

EQUATIONS

  OBJ            Objective function - input orientation

  OUTPUT(mm)     Expansion of DMUs’ outputs

  UCINPUT(ucnn)  Constraint of DMUs’ uncontrollable inputs

  INPUT(cnn)     Contraction of DMUs’ controllable inputs

  VRS            Constraint for variable returns to scale

  DRS            Constraint for non-increasing returns to scale

*                DRS constraint not used in the rail study

;

*********************************************************************************

* The following statements define the objective and constraint equations of the *

* DEA model as given in Banker, R.D. and R.C. Morey (1986) “Efficiency Analysis *

* for Exogenously Fixed Inputs and Outputs”, Operations Research, vol. 34,      *

* no. 4, pp. 513 – 521.                                                         *

* Model specifications are selected later in a loop of program statements,      *

* which also compute the model solutions.                                       *

*********************************************************************************

OBJ..            Z =E= PHI;

OUTPUT(mm)..     SUM((i,y)$catgroup(i,y),

                                       DMUYY(i,y,mm)*LAMBDA(i,y)) =G= DMUOUT(mm);

INPUT(cnn)..     PHI * DMUINP(cnn)- SUM((i,y)$catgroup(i,y),

                                      DMUXX(i,y,cnn)*LAMBDA(i,y)) =G= 0;

UCINPUT(ucnn)..  DMUINP(ucnn) - SUM((i,y)$catgroup(i,y),

                                     DMUXX(i,y,ucnn)*LAMBDA(i,y)) =G= 0;

VRS ..           SUM((i,y)$catgroup(i,y),            LAMBDA(i,y)) =E= 1;

DRS ..           SUM((i,y)$catgroup(i,y),            LAMBDA(i,y)) =L= 1;

*********************************************************************************

* The following ‘model’ statements select equations from the above list to      *

* specify the CRS and VRS models for estimating productivity and technical      *

* efficiency, respectively. A third alternative assumption of DRS is not used   *

* in the study.                                                                 * 

*********************************************************************************

MODEL DMUCRS  / UCINPUT, INPUT, OUTPUT, OBJ      /

      DMUVRS  / UCINPUT, INPUT, OUTPUT, OBJ, VRS /

;

DMUCRS.workspace  = 1.10;

DMUVRS.workspace  = 1.10;

OPTION  PHIHAT:3:2:1;

OPTION  PEERHAT:3:2:2;

OPTION  OUTSLACK:3:2:1;

OPTION  INPEXCES:3:2:1;

PUT DEARESULT;

PUTHD 'Rail study' ///;

   DMUXX(i,y,nn)  = (TINY+ DMUX(i,y,nn)); 

   DMUYY(i,y,mm)  = (TINY+ DMUY(i,y,mm));

*********************************************************************************

* The following ‘loop’ statements select observations from the sample list      *

* for inclusion in the model. The selection criteria are based on the           *

* completeness of input and output data, the chosen country coverage, study     *

* period (or window width in the case of window analysis), and categorical data.*

*********************************************************************************

ITER3 = 1 ; 

LOOP(iter1 $ (ITER3 < ITERNO + 1),   

   ITER2=1;

      window(y)= yes;

      LOOP(y, 

      window(y)  $ ( ITER2 < ITER3 or ITER2 > WINWIDTH+ITER3-1 )= NO ;

      ITER2 = ITER2 + 1;

      );

   sample(i,y) = iy(i,y) $ (coverage(i) and window(y) ) ;

PUT 'DEA Calculation iteration no.' iter3::0 /   ;

    LOOP((ip,yp) $ sample (ip,yp), 

    LOOP((ipp,ypp),

    catgroup(ipp,ypp) = NO;

    catgroup(ipp,ypp) = sample(ipp,ypp) $ (DMUCV(ipp,ypp) <= DMUCV(ip,yp));

  ) ;

*********************************************************************************

* The program can differentiate the performance of observations that are deemed *

* to be efficient. This method is laid out in Andersen, P. and N.C. Petersen    *

* (1993)“A Procedure for Ranking Efficient Units in Data Envelopment Analysis”, *

* Management Science, vol. 39, no. 10, pp. 1261 – 1264.                         *

* This facility is not used in the rail study, and is disabled by excluding the *

* following statement from the program code.                                    *

*********************************************************************************

*  catgroup(ip,yp) = NO ;

    DMUINP(nn)     = (TINY+ DMUX(ip,yp,nn)    );

    DMUOUT(mm)     = (TINY+ DMUY(ip,yp,mm)    );

*********************************************************************************

* The minimisation problem is solved for each included observation separately   *

* and linear programming techniques.                                            *

*********************************************************************************

SOLVE DMUCRS USING LP MINIMIZING Z;

*********************************************************************************

* The program code below is to store the model solutions in a csv format for    *

* further processing.                                                           *

*********************************************************************************

        PHIHAT(ip,yp,'SCORE')                      = Z.L;

        PEERHAT('WEIGHTS',ip,yp,i,y)$catgroup(i,y) = LAMBDA.L(i,y);

        OUTSLACK('SLACK',ip,yp,mm)$(DMUOUT(mm) > TINY )

      = 100*(SUM((i,y)$catgroup(i,y),

        DMUYY(i,y,mm)*LAMBDA.L(i,y))/DMUOUT(mm)-1);

        INPEXCES('SLACK',ip,yp,ucnn)$(DMUINP(ucnn) > TINY  )

      = 100*(DMUINP(ucnn) - SUM((i,y)$catgroup(i,y),

        DMUXX(i,y,ucnn)*LAMBDA.L(i,y)))/DMUINP(ucnn);

        INPEXCES('SLACK',ip,yp,cnn)$(DMUINP(cnn) > TINY  )

      = 100*(Z.L * DMUINP(cnn) - SUM((i,y)$catgroup(i,y),

        DMUXX(i,y,cnn)*LAMBDA.L(i,y)))/DMUINP(cnn);

) ;

DISPLAY PHIHAT, PEERHAT;

  PUT / 'Efficiency scores and slack variables for DMUs' /

        'included in this calculation are:' /; 

  PUT   '       ', 'SCORES' ;

  LOOP(mm, 

        PUT mm.te(mm):>8  

  );

  LOOP(nn, 

        PUT nn.te(nn):>8  

  );

  PUT /;

  LOOP ((i,y)$sample(i,y), 

        PUT sample.te(i,y):12, PHIHAT(i,y,'SCORE'):12:3 ; 

        LOOP (mm, 

        PUT OUTSLACK('SLACK',i,y,mm) ;    

        ); 

        LOOP (nn, 

        PUT INPEXCES('SLACK',i,y,nn) ;    

        ); 

        PUT / ;

  );      

PUT // 'For each of the DMUs listed in the column, ' /

         'its benchmarking comparators are:' // ;

  PUT '          ';

  LOOP((i,y)$(sample(i,y) and sum((ip,yp), PEERHAT('WEIGHTS',ip,yp,i,y)) > 0), 

        PUT sample.te(i,y):>12  ;

  );

  PUT /;

  LOOP ((ip,yp)$sample(ip,yp), 

      PUT sample.te(ip,yp):<12  ;

      LOOP((i,y)$(sample(i,y) and sum((ipp,ypp),

        PEERHAT('WEIGHTS',ipp,ypp,i,y)) > 0),

  PUT PEERHAT('WEIGHTS',ip,yp,i,y):>12:3  ; 

      );

  PUT /;

  );

  PHIHAT(i,y,'SCORE')           = 0.0; 

  PEERHAT('WEIGHTS',ip,yp,i,y)  = 0.0;

  OUTSLACK('SLACK',ip,yp,mm)    = 0.0;

  INPEXCES('SLACK',ip,yp,nn)    = 0.0;

ITER3 = ITER3 + 1; 

);

